
Σκοπός του εγχειριδίου 
 

Σκοπός του παρόντος εγχειριδίου είναι να παρέχει κατάλληλες 
πληροφορίες στον εκπαιδευτικό, ερευνητή ή προχωρηµένο χρήστη, που 
επιθυµεί να τροποποιήσει τις λειτουργικότητες της γεννήτριας παιχνιδιών  
Sus-X, αλλάζοντας τα υπάρχοντα χαρακτηριστικά της ή και προσθέτοντας 
νέες δυνατότητες.  

Για το σκοπό αυτό καταβάλλεται προσπάθεια ώστε να γίνουν 
κατανοητές οι διεργασίες που πραγµατοποιούνται κατά το παίξιµο του 
παιχνιδιού, ώστε να είναι δυνατή η αποδόµησή του.  
 

Σχεδιάζοντας ένα νέο παιχνίδι 
 

Κατά το σχεδιασµό-δηµιουργία ενός νέου παιχνιδιού δεν 
χρησιµοποιείται προγραµµατιστικός κώδικας (logo) παρά µόνο απλές εντολές 
εµφάνισης «Όψεων» οι οποίες είναι τοποθετηµένες µέσα στα «γεγονότα» των 
κουµπιών.  

Οι «Όψεις» είναι εγγενής λειτουργικότητα της συγγραφικής 
πλατφόρµας «Αβάκιο» και αποτελούν αποθηκευµένες καταστάσεις εµφάνισης 
των ψηφίδων στην οθόνη. Με τις «Όψεις» µπορούµε να τοποθετούµε, 
εµφανίζουµε και να αποκρύπτουµε ψηφίδες και να αποθηκεύουµε (µε όνοµα 
της επιλογής µας) την ακριβή εµφάνιση του παιχνιδιού (ή µικρόκοσµου 
γενικότερα). 
 

Παίζοντας το παιχνίδι 
 
Μετά τη διαδικασία σχεδιασµού, όταν πλέον είναι έτοιµο το νέο παιχνίδι, όλες 
οι διεργασίες πραγµατοποιούνται µέσω προγράµµατος σε γλώσσα logo το 
οποίο είναι καταχωρηµένο στην ψηφίδα “Logo”. 

 

Έναρξη παιχνιδιού 
 
Κατά την έναρξη του παιχνιδιού (κουµπί «¨Εναρξη»): 

• Ξεκινά το χρονόµετρο να µετρά το χρόνο 
• Φορτώνονται οι αρχικές τιµές στις ιδιότητες του παιχνιδιού 
• Αρχικοποιείται η µέτρηση του πλήθους των σηµείων 
• Εµφανίζεται το κουµπί «Επιλογή σηµείου» για την επιλογή των 

σηµείων 
• Αλλάζει εµφάνιση το κουµπί «Έναρξη» 
• Καθαρίζεται η Β∆ του παιχνιδιού από στοιχεία προηγούµενων 

παιχνιδιών 
• Ξεκινάει η καταγραφή των κινήσεων του παίκτη (logging) 

Επιλογή σηµείου από τον χάρτη  
 
Κατά την επιλογή ενός σηµείου (στο χάρτη) από τον παίκτη (κουµπί «Επιλογή 
σηµείου»): 
 

• ∆ιαγράφεται οποιοδήποτε παλαιότερο µήνυµα από το πλαίσιο 
κειµένου που εµφανίζει πληροφορίες 

• Αυξάνει το πλήθος των επιλεγµένων σηµείων 
• Προσθέτει µια νέα εγγραφή στη Β∆ του παιχνιδιού µε όλα τα 

στοιχεία – ιδιότητες του επιλεγµένου σηµείου 
• Ελέγχεται εάν συµπληρώθηκε ο µέγιστος αριθµός σηµείων 
• Ελέγχεται εάν κάποια από τις τιµές των ιδιοτήτων ενεργοποιεί 

κάποιο από τα κριτήρια τερµατισµού που έχει θέσει ο σχεδιαστής 
• Εάν δεν υπάρχει κάποια αιτία τερµατισµού του παιχνιδιού, 

ελέγχονται οι «συνθήκες ελέγχου» που έχει θέσει ο σχεδιαστής και 
αν κάποια από αυτές ενεργοποιείται, εµφανίζεται το καταχωρηµένο 
µήνυµα. 

• Πραγµατοποιείται καταγραφή της ενέργειας του χρήστη (διαδικασία 
logrecord) 

 

 

Τερµατισµός παιχνιδιού 
 

Αν το παιχνίδι τερµατιστεί από «εσωτερική» αιτία, δηλαδή επειδή έληξε 
ο χρόνος, ή συµπληρώθηκε το πλήθος των σηµείων ή ενεργοποιήθηκε µια 
συνθήκη- κριτήριο τερµατισµού του σχεδιαστή : 

• Σταµατά το χρονόµετρο 
• Εµφανίζονται πληροφορίες για την κατάσταση του παιχνιδιού τη 

στιγµή του τερµατισµού 
• Εµφανίζεται περιγραφή της αιτίας τερµατισµού 
• Αλλάζει εµφάνιση το κουµπί «∆ιακοπή» 
• Αποκρύπτεται το κουµπί «Επιλογή σηµείου» 
 
Αν το παιχνίδι τερµατιστεί από τον παίκτη (κουµπί «∆ιακοπή») 

πραγµατοποιούνται οι ίδιες (παραπάνω) ενέργειες αλλά δεν εµφανίζεται 
µήνυµα µε την περιγραφή της αιτίας τερµατισµού. 

Συνοπτική περιγραφή των διαδικασιών logo 
 
 
 
 
Έναρξη ή Τερµατισµός παιχνιδιού  

• Έλεγχος αν ζητήθηκε έναρξη ή τερµατισµός (από το ίδιο κουµπί) 
• Ανάκτηση από τη Β∆  “Game_properties”  των αρχικών τιµών για την έναρξη του παιχνιδιού 
• ∆ιαγραφή των στοιχείων (από προηγούµενο παιχνίδι) στη Β∆ του χρήστη (game_data) 

 
to New_Game ; when start or stop game button is pressed 
localmake "action ask "new [button.text] 
if :action="|Stop Game| [ 
ask "chrono [stopchronometer] 
ask "new [button.settext "|Start Game|] 
ask "new [button.setfgcolor [0 150 0]] 
ask "accept [hide] 
ask "|Remaining| [hide] 
stop 
] 
make "MaxTurns ask "Turns [sbutton.value] 
make "TimeLimit ask "Available_time [sbutton.value] 
make "EndReason "none 
make "FieldNames ask "map_data [db.fieldnames] 
make "FieldNames Butfirst :Fieldnames 
make "FieldNames Butfirst :Fieldnames 
make "FieldCount length :fieldnames 
make "FieldCount :FieldCount 
ask "chrono [resetchronometer] 
ask "chrono [startchronometer] 
ask "Available_time [hide] 
Ask "Turns [hide] 
ask "lbl_time [hide] 
ask "lbl_turns [hide] 
ask "TXT_end [area.settext "||] 
ask "TXT_Check [area.settext "||] 
ask "TXT_end [hide] 
ask "TXT_check [hide] 
Ask "Accept [restore] 
ask "new [button.settext "|Stop Game|] 
ask "new [button.setfgcolor [255 0 0]] 
ask "|Remaining| [restore] 
localmake "recordcount ask "game_data [db.recordcount] 
repeat :recordcount [ask "game_data [db.removerecord 1 ]] 
ask "Game_data [db.addrecord] 
repeat :FieldCount [ 
     localmake "initValue ask "Game_properties [db.cell repcount "Initial_value] 
        localmake "curFieldcode ask "Game_properties [db.cell repcount "Field_code] 
 ask "Game_data [db.setcell 1 :curFieldcode  :initValue]  
] 
make "curPoint 0 
make "TurnsRemain :Maxturns 
ask "logtext [area.settext "||] 
logrecord true 

end 

 
 
 
Επιλογή - αποδοχή του επιλεγµένου σηµείου του χάρτη 

• Ανάκτηση των τιµών του σηµείου από τη Β∆ του χάρτη και εισαγωγή τους στη Β∆ του 
χρήστη (game_data) 

• Έλεγχος αν συµπληρώθηκε ο µέγιστος αριθµός σηµείων – Αν συµπληρώθηκε, τερµατισµός 
του παιχνιδιού (κλήση της GameEnded) 

• Έλεγχος αν το παιχνίδι πρέπει να τερµατιστεί λόγω επιλεγµένων συνθηκών 
(CheckEndConditions) 

• Έλεγχος για τις τιµές των ιδιοτήτων µε σκοπό την εµφάνιση προειδοποιητικών µηνυµάτων 
(CheckConditions) 

to Accept_point ;get and save the chosen point 
ask "TXT_check [area.settext "||] 
make "curPoint :curPoint+1 
make "messsageStr "|| 
ask "game_data [db.addrecord] 
localmake "curMapRecord ask "map_data [db.activerecord] 
repeat :FieldCount [ 
        localmake "curMapField item repcount :FieldNames 
 localmake "curFieldValue ask "map_data [db.cell :curMapRecord :curMapField] 
 localmake "curField word "Field repcount 
 localmake "curRec :curPoint+1 
 ask "game_data [db.setcell :curRec :curField :curFieldValue] 
] 
If :curPoint = :MaxTurns [ 
 make "EndReason "byMaxTurns 
        GameEnded 
 ] 
CheckEndConditions 
if :EndReason="none [ 
   checkconditions 
   GetAndShowValues "TXT_check 
] 
make "TurnsRemain :MaxTurns - :curPoint 
logrecord false 
end 
 
 
 
Έλεγχος ιδιοτήτων για την εµφάνιση προειδοποιητικών µηνυµάτων 

• Ανάκτηση καταχωρηµένων συνθηκών ελέγχου από την Β∆ «Check_conditions» 
• Σύγκριση ιδιοτήτων µε τις συνθήκες ελέγχου και εµφάνιση συγκεντρωτικού µηνύµατος  

to CheckConditions ;check conditions after every point acceptance 
localmake "numofConditions ask "Check_conditions [db.recordcount] 
repeat :numofconditions [ 
   localmake "IsEnabled ask "Check_conditions [db.cell repcount "Con_Enabled] 
   if :IsEnabled=true [ 
    localmake "curFieldCheck ask "Check_conditions [db.cell repcount "Field_Name] 
    localmake "curConOperand ask "Check_conditions [db.cell repcount "Con_Operand] 
   localmake "curConValue ask "Check_conditions [db.cell repcount "Con_Value] 
    localmake "curConmessage ask "Check_conditions [db.cell repcount "Con_Message] 
   localmake "curConmessage word :curConmessage "| 
| 
        FieldNameToFieldCode :curFieldCheck 
    localmake "curFieldValue ask "game_data [db.dsum :curFieldcode] 
   if :curConOperand = "= [ 
        if :curFieldValue = :curConValue [ ask "TXT_check [area.append :curConmessage]]  
          ] 
    if :curConOperand = "> [ 
        if :curFieldValue > :curConValue [ ask "TXT_check [area.append :curConmessage]]  
          ] 
    if :curConOperand = "< [ 
        if :curFieldValue < :curConValue [ ask "TXT_check [area.append :curConmessage]]  
          ] 
  ] 
] 
ask "TXT_check [restore] 
end 
 
 
Έλεγχος ικανοποίησης συνθηκών τερµατισµού του παιχνιδιού 

• Ανάκτηση καταχωρηµένων συνθηκών τερµατισµού από την Β∆ «End_conditions» 
• Σύγκριση ιδιοτήτων µε τις συνθήκες τερµατισµού  

 
to CheckEndConditions ;check if the game is ended by end conditions 
localmake "numofEndConditions ask "End_conditions [db.recordcount] 
repeat :numofendconditions [ 
   localmake "IsEnabled ask "End_conditions [db.cell repcount "End_Enabled] 
   if :IsEnabled=true [ 
    localmake "curFieldCheck ask "End_conditions [db.cell repcount "Field_Name] 
    localmake "curEndOperand ask "End_conditions [db.cell repcount "End_Operand] 
   localmake "curEndValue ask "End_conditions [db.cell repcount "End_Value] 
    localmake "curEndMessage ask "End_conditions [db.cell repcount "End_Message] 
        localmake "curEndMessage word :curEndMEssage "| 
| 
   FieldNameToFieldCode :curFieldCheck 
    localmake "curFieldValue ask "game_data [db.dsum :curFieldcode] 
   if :curEndOperand = "= [ 
        if :curFieldValue = :curEndValue [  
   ask "TXT_End [area.append :curEndMessage] 
   make "EndReason "byConditions   
   ]]  
        if :curEndOperand = "> [ 
        if :curFieldValue > :curEndValue [  
   ask "TXT_End [area.append :curEndMessage] 
   make "EndReason "byConditions 
   ]]  
    if :curEndOperand = "< [ 
        if :curFieldValue < :curEndValue [  
   ask "TXT_End [area.append :curEndMessage] 
   make "EndReason "byConditions 

   ]] 
  ] 
] 
if :EndReason="byConditions [GameEnded] 
end 

 
Υπολογισµός και εµφάνιση ιδιοτήτων (των τιµών τους) 

• Υπολογισµός τιµών ιδιοτήτων στη Β∆ του χρήστη (game_data) 

• Σύνθεση και εµφάνιση κειµένου µε όλες τις πληροφορίες 

to GetAndShowValues :TextAreaName ; Calculates and shows all field values 
  repeat 10 [ 
 Localmake "curFieldCode word "Field repcount 
 Localmake "curFieldValue ask "game_data [db.dsum word "Field repcount] 
FieldCodeToFieldName :curFieldCode 
        print :curFieldCode 
 ifelse :curFieldName ="||[][ 
     localmake "lineText word :curFieldName "= 
  localmake "lineText word :lineText :curFieldValue 
  ask :TextAreaName [area.append :lineText] 
  ask :TextAreaName [area.append "|   
|] 
 ] 
     ] 
end 

 
 
 
Έλεγχος χρονικού ορίου παιχνιδιού 

• Υπολογισµός διάρκειας παιχνιδιού 

• Εµφάνιση υπολειπόµενου χρόνου 

to CheckTimeLimit ; checks if time is up 
localmake "curTime chronometerminutes  
ifelse :curTime > :timelimit [ 
        make "EndReason "byTime         
        GameEnded 
][ 
make "timeRemain :timelimit - integer :curTime 
localmake "lblRemaining word "|Remaining...| :timeRemain 
localmake "lblRemaining word :lblRemaining "| minute(s) -- | 
localmake "lblRemaining word :lblRemaining :TurnsRemain 
localmake "lblRemaining word :lblRemaining "| turn(s)| 
] 
Ask "Remaining [label.settext :lblRemaining] 
end 
 
 
∆ιαδικασία Τερµατισµού Παιχνιδιού 

• ∆ιακοπή χρονοµέτρου 
• Έλεγχος αιτίας τερµατισµού  -  εµφάνιση κατάλληλου µηνύµατος 
• Εµφάνιση ιδιοτήτων µε κλήση της διαδικασίας GetAndShowValues 

to GameEnded ;shows proper message when game is ended 
ask "chrono [stopchronometer] 
ask "|Remaining| [hide] 
if :EndReason="byTime [make "messageStr "|The Time Is Up!|] 
if :EndReason="byConditions [make "messageStr "|The Game is Ended!|] 
if :EndReason="byMaxTurns [make "messageStr "|You have reached the maximum number of turns|] 
make "messageStr word :messageStr "| 
| 
ask "TXT_end [area.append :messageSTR] 
ask "TXT_end [restore] 
ask "TXT_check [hide] 
ifelse and :EndReason="byConditions :curPoint=:Maxturns [][ 
GetAndShowValues "TXT_end] 
Ask "Accept [hide] 
ask "new [button.settext "|Start Game|] 
ask "new [button.setfgcolor [0 150 0]] 
end 
 
 
Βοηθητικές διαδικασίες 

• FieldCodeToFieldName -> δέχεται τον αύξοντα αριθµό της ιδιότητας (field) και επιστρέφει το 
όνοµα της 

• FieldNameToFieldCode -> δέχεται το όνοµα της ιδιότητας (field) και επιστρέφει τον αύξοντα 
αριθµό της 

• Και στις δύο περιπτώσεις εκτελείται ερώτηµα (query) στη Β∆ ρυθµίσεων του παιχνιδιού 
(game_properties) 

to FieldCodeToFieldName :fcode ;Gets Field code (ie Field1) -->returns the Field name 
localmake "query word "Field_code= :fcode 
localmake "selected ask "Game_properties [db.select :query ] 
localmake "selected item 1 :selected ;convert to integer 
make "curFieldName ask "game_properties [db.cell :selected "Field_name ] 
end 
to FieldNameToFieldCode :fname ;Gets Field Name -->returns the Field code 
print :fname 
localmake "query word "Field_name= :fname 
print :query 
localmake "selected ask "Game_properties [db.select :query ] 
print :selected 
localmake "selected item 1 :selected ;convert to integer 
make "curFieldCode ask "game_properties [db.cell :selected "Field_code ] 
end 
 
Καταγραφή ενεργειών (δηµιουργία εγγραφών logging) 

• Καλείται κατά την έναρξη του παιχνιδιού, κατά την αποδοχή σηµείου στο χάρτη και κατά το 
τέλος του παιχνιδιού 

• Καταγράφει τη χρονική στιγµή και τις τιµές των ιδιοτήτων 

to LogRecord :newgame 
ifelse :newgame = true [ 
  localmake "curDescr "|NEW GAME| 
  ][ 
  localmake "curdbaserecord ask "map_data [ db.activerecord] 
  localmake "curDescr ask "map_data [db.cell :curdbaserecord "Description ] 
] 
make "curTime (chronometertime) 
localmake "timeword item 1 :curtime 
localmake "timeword word :timeword ": 
localmake "timeword word :timeword item 2 :curtime 
localmake "timeword word :timeword ": 
localmake "timeword word :timeword item 3 :curtime 
localmake "linetext word :curPoint + 1 "|# | 
localmake "linetext word :lineText :timeword 
localmake "linetext word :lineText "| [| 
localmake "linetext word :lineText :curDescr 
localmake "linetext word :lineText "|]| 
repeat :FieldCount [ 
 FieldCodeToFieldName word "Field repcount 
 localmake "linetext word :lineText "| | 
 localmake "lineText word :lineText :curFieldName 
 localmake "linetext word :lineText "= 
 ifelse :newgame=true [ 
    localmake "curFieldVal ask "game_data [db.cell :curpoint + 1 word "Field repcount] 
 ][ 
    localmake "curFieldVal ask "game_data [db.dsum word "Field repcount] 
 ] 
 localmake "lineText word :lineText :curFieldVal   
] 
ask "logtext [area.append :linetext] 
ask "logtext [area.append "| 
|] 
end 
 


